StikeDipRakeBox.nb (formerly BrickCode.nb)
This illustrates the distance between double couple moment tensors in the strike-dip-rake box. The moral is that coordinates that are far from each other in the strike-dip-rake box can be very close to each other in the space of moment tensors.

Run common_funs.nb, then commonMT_funs.nb
κ : strike angle [0, 360]
θ : dip angle [0, 90] (the code uses math convention ϕ)
σ : rake angle [-90, 90]

Note: All angles are stored in radians.
Note: In the code, the angles are ordered as strike-rake-dip (following Tape and Tape, 2012), but in the printed text (and in the notebook name), the order is strike-dip-rake.

Moment tensor from eigenvalues and strike-dip-rake angles

In[2501]:=
```
MT[Λ_, {κ_, σ_, ϕ_}] := UgHat[{κ, σ, ϕ}].DiagonalMatrix[Λ].Transpose[UgHat[{κ, σ, ϕ}]];
```

Choose a distance function
note: the distances are scaled to the range [0, 1]

In[2500]:=
```
UseAngularDist = False;
```

In[2502]:=
```
(*GraphicsRow[Style[Text[#],Background→Hue[#]]&/@Range[0,1,0.1]]
 Hue[0.72]*)
```

In[2503]:=
```
MTdistance[Λ1_, Λ2_, {κ1_, σ1_, ϕ1_}, {κ2_, σ2_, ϕ2_}] :=
  NormMatrix[MT[Λ2, {κ2, σ2, ϕ2}] - MT[Λ1, {κ1, σ1, ϕ1}]];
If[UseAngularDist,
  MTDCdistance[{κ1_, σ1_, ϕ1_}, {κ2_, σ2_, ϕ2_}] :=
   1/π AngleMatrix[MT[unit[{1, 0, -1}], {κ1, σ1, ϕ1}], MT[unit[{1, 0, -1}], {κ2, σ2, ϕ2}]],
  MTDCdistance[{κ1_, σ1_, ϕ1_}, {κ2_, σ2_, ϕ2_}] :=
   1/2 MTdistance[unit[{1, 0, -1}], unit[{1, 0, -1}], {κ1, σ1, ϕ1}, {κ2, σ2, ϕ2}]];
```

In[2505]:=
```
HueMtDistTo[{κ_, σ_, ϕ_}, {κ0_, σ0_, ϕ0_}] :=
  Hue[0.72 (1 - MTDCdistance[{κ, σ, ϕ}, {κ0, σ0, ϕ0}])];
```

Plotting...

In[2506]:=
```
space = .1; del = .05 π; (* small del gives better resolution *)
```

In[2507]:=

```
MoveRearPoly[poly_] := (1  0
                        0 -1).# + {0, π + space} & /@ poly
MoveFrontPoly[poly_] := # + {0, -π - space} & /@ poly
MoveRightPoly[poly_] := Rot2D[π / 2].# + {5 π / 2 + space, 0} & /@ poly
MoveLeftPoly[poly_] := (-1 0
                         0 1).Rot2D[π / 2].# + {-π / 2 - space, 0} & /@ poly
MoveTopPoly[poly_] := (1  0
                       0 -1).# + {0, 3 π / 2 + 2 space} & /@ poly
```

In[2512]:=

```
FoldUpBox[{κ0_, σ0_, ϕ0_}] := Graphics[{
   Arrow[{{0, 2 π + 4 space}, {2 π, 2 π + 4 space}}],   (* for κ *)
   Arrow[{{5 π / 2 + 3 space, -π / 2}, {5 π / 2 + 3 space, π / 2}}],     (* for σ *)
   Arrow[{{2 π + 2 space, 2 π + 2 space}, {2 π + 2 space, π + 2 space}}],     (* for σ *)
   Arrow[{{2 π + 2 space, π + space}, {2 π + 2 space, π / 2 + space}}],  (* for dip angle *)
   Arrow[{{2 π + 2 space, -π - space}, {2 π + 2 space, -π / 2 - space}}],  (* for dip angle *)
   Arrow[{{2 π + π / 2 + space, -π / 2 - 2 space}, {2 π + space, -π / 2 - 2 space}}],
   (* for dip angle on right end flap *)
   Text[StyleForm["κ", FontSize → 18], {π, 2 π + 6 space}],
   Text[StyleForm["σ", FontSize → 18], {5 π / 2 + 6 space, 0}],
   Text[StyleForm["σ", FontSize → 18], {2 π + 6 space, 3 π / 2 + 2 space}],
   Text[StyleForm["θ", FontSize → 18], {2 π + 5 space,  π / 2 + space + π / 4}],
   Text[StyleForm["θ", FontSize → 18], {2 π + 5 space, -π / 2 - 3 space - π / 4}],
   Text[StyleForm["θ", FontSize → 18], {2 π + π / 4 + 3 space, -π / 2 - 6 space}],
   (* for dip angle on right end flap *)
   {HueMtDistTo[{Mean[#]〚1〛, Mean[#]〚2〛,  π / 2}, {κ0, σ0, ϕ0}], Polygon[#]} & /@
    BasicPolys[Range[0, 2 π, del], Range[-π / 2, π / 2, del]], (*bottom*)
   {HueMtDistTo[{Mean[#]〚1〛,  π / 2, Mean[#]〚2〛}, {κ0, σ0, ϕ0}],
      Polygon[MoveRearPoly[#]]} & /@
    BasicPolys[Range[0, 2 π, del], Range[0, π / 2, del]], (*rear*)
   {HueMtDistTo[{Mean[#]〚1〛, -π / 2, Mean[#]〚2〛}, {κ0, σ0, ϕ0}],
      Polygon[MoveFrontPoly[#]]} & /@
    BasicPolys[Range[0, 2 π, del], Range[0, π / 2, del]], (*front*)
   {HueMtDistTo[{   2 π, Mean[#]〚1〛, Mean[#]〚2〛}, {κ0, σ0, ϕ0}],
      Polygon[MoveRightPoly[#]]} & /@
    BasicPolys[Range[-π / 2, π / 2, del], Range[0, π / 2, del]],    (*right*)
   {HueMtDistTo[{    0, Mean[#]〚1〛, Mean[#]〚2〛}, {κ0, σ0, ϕ0}],
      Polygon[MoveLeftPoly[ #]]} & /@
    BasicPolys[Range[-π / 2, π / 2, del], Range[0, π / 2, del]],   (*left*)
   {HueMtDistTo[{   Mean[#]〚1〛, Mean[#]〚2〛,   0}, {κ0, σ0, ϕ0}],
      Polygon[MoveTopPoly[ #]]} & /@
    BasicPolys[Range[0, 2 π, del], Range[-π / 2, π / 2, del]] (*top*)
  }, AspectRatio → Automatic, AxesLabel → {"κ", "σ"}, Axes → False,
  Ticks → {Range[0, 2 π, π / 3], Range[-π / 2, π / 2, π / 4]}, AspectRatio → Automatic]
```

In[2513]:=

```
TopBrickFace = {{0, -π / 2, π / 2}, {2 π, -π / 2, π / 2}, {2 π, π / 2, π / 2}, {0, π / 2, π / 2}};
(* for κσϕ, other file has for κσh *)
BottomBrickFace = {{0, -π / 2, 0}, {2 π, -π / 2, 0}, {2 π, π / 2, 0}, {0, π / 2, 0}};
FrontBrickFace = {{0, -π / 2, 0}, {2 π, -π / 2, 0}, {2 π, -π / 2, π / 2}, {0, -π / 2, π / 2}};
BackBrickFace = {{0,  π / 2, 0}, {2 π,  π / 2, 0}, {2 π,  π / 2, π / 2}, {0,  π / 2, π / 2}};
RightBrickFace = {{2 π,  -π / 2, 0}, {2 π,  π / 2, 0}, {2 π,  π / 2, π / 2}, {2 π,  - π / 2, π / 2}} ;
LeftBrickFace = {{0,  -π / 2, 0}, {0,  π / 2, 0}, {0,  π / 2, π / 2}, {0,  - π / 2, π / 2}};
```

In[2519]:=

```
GreenBoxFaces = {          (* for κσϕ, other file has for κσh *)
   TopBrickFace, BottomBrickFace,
   FrontBrickFace, BackBrickFace, RightBrickFace, LeftBrickFace};
```

In[2520]:=

```
PolyToFront[poly_] := {#〚1〛, -π / 2, π / 2 - #〚2〛} & /@ poly;
PolyToRear[poly_] := {#〚1〛, π / 2, π / 2 - #〚2〛} & /@ poly;
PolyToBottom[poly_] := {#〚1〛, #〚2〛, 0} & /@ poly;
PolyToTop[poly_]    := {#〚1〛, #〚2〛, π / 2} & /@ poly;
PolyToϕ0[poly_, ϕ0_]    := {#〚1〛, #〚2〛, ϕ0} & /@ poly;
PolyToLeft[poly_] := {0, #〚1〛, π / 2 - #〚2〛} & /@ poly;
PolyToLRight[poly_] := {2 π, #〚1〛, π / 2 - #〚2〛} & /@ poly;
```

In[2527]:=

```
BoxIn3D[{κ0_, σ0_, ϕ0_}, WantTop_, WantFront_, WantRight_] := Graphics3D[{
   {HueMtDistTo[{Mean[#]〚1〛, Mean[#]〚2〛,  π / 2}, {κ0, σ0, ϕ0}],
      EdgeForm[], Polygon[PolyToBottom[#]]} & /@
    BasicPolys[Range[0, 2 π, del], Range[-π / 2, π / 2, del]], (*bottom*)
   {HueMtDistTo[{Mean[#]〚1〛,  π / 2, Mean[#]〚2〛}, {κ0, σ0, ϕ0}],
      EdgeForm[], Polygon[PolyToRear[#]]} & /@
    BasicPolys[Range[0, 2 π, del], Range[0, π / 2, del]], (*rear*)
   {HueMtDistTo[{    0, Mean[#]〚1〛, Mean[#]〚2〛}, {κ0, σ0, ϕ0}],
      EdgeForm[], Polygon[PolyToLeft[ #]]} & /@
    BasicPolys[Range[-π / 2, π / 2, del], Range[0, π / 2, del]],   (*left*)
   If[WantFront == "WantFront",
    {{HueMtDistTo[{Mean[#]〚1〛, -π / 2, Mean[#]〚2〛}, {κ0, σ0, ϕ0}], EdgeForm[],
        Polygon[PolyToFront[#]]} & /@ BasicPolys[Range[0, 2 π, del], Range[0, π / 2, del]],
     {FaceForm[], Polygon[FrontBrickFace]}}, {}], (*front*)
   If[WantRight == "WantRight",
    {{HueMtDistTo[{   2 π, Mean[#]〚1〛, Mean[#]〚2〛}, {κ0, σ0, ϕ0}], EdgeForm[], Polygon[
         PolyToLRight[#]]} & /@ BasicPolys[Range[-π / 2, π / 2, del], Range[0, π / 2, del]],
     {FaceForm[], Polygon[RightBrickFace]}}, {}],    (*right*)
   If[WantTop == "WantTop",
    {{HueMtDistTo[{   Mean[#]〚1〛, Mean[#]〚2〛,   0}, {κ0, σ0, ϕ0}], EdgeForm[],
        Polygon[PolyToTop[ #]]} & /@ BasicPolys[Range[0, 2 π, del], Range[-π / 2, π / 2, del]],
     {FaceForm[], Polygon[TopBrickFace]}}, {}], (*top*)
   {FaceForm[], Polygon[#]} & /@ {BottomBrickFace, LeftBrickFace, BackBrickFace}},
  Lighting → {{"Ambient", White}}, Boxed → False, ViewPoint → 10 xyztp[{-60, 50}]]
```

Optional checks

example strike-rake-dip triple (in radians)

In[2528]:=

```
{κ0, σ0, ϕ0} = {π/3, π/3, π/3};
```

find a/the point in the box that is the maximal MT distance from the target MT

In[2529]:=

```
Clear[κ, σ, ϕ]
MaxDist[{κ0_, σ0_, ϕ0_}] := Module[{solution, ωmax, κmax, σmax, ϕmax},
  solution = NMaximize[{MTDCdistance[{κ0, σ0, ϕ0}, {κ, σ, ϕ}],
     0 ≤ κ ≤ 2 π && -π / 2 ≤ σ ≤ π / 2 && 0 ≤ ϕ ≤ π / 2}, {κ, σ, ϕ}];
  {ωmax, vars} = solution;
  {κmax, σmax, ϕmax} = {κ, σ, ϕ} /. vars]
```

In[2531]:=

```
{κmax, σmax, ϕmax} = MaxDist[{κ0, σ0, ϕ0}];
{κmax, σmax, ϕmax} / Degree
N[{κ0, σ0, ϕ0} / Degree]
MTDCdistance[{κ0, σ0, ϕ0}, {κmax, σmax, ϕmax}]
```

Out[2532]=

```
{289.107, -49.1066, 41.4096}
```

Out[2533]=

```
{60., 60., 60.}
```

Out[2534]=

```
1.
```

check:
The farthest-away MT should be -M0.
The distance from M0 to -M0 is 180 and red.
The distance from M0 to M0 is 0 and blue.

In[2535]:=

```
Mmax = MT[{1, 0, -1}, {κmax, σmax, ϕmax}];
MatrixForm[Mmax + M0]
```

Out[2536]//MatrixForm=

$$\begin{pmatrix} -3.66415\times10^{-9} & 7.69065\times10^{-9} & 2.68725\times10^{-9} \\ 7.69065\times10^{-9} & 1.4514\times10^{-9} & 2.72295\times10^{-9} \\ 2.68725\times10^{-9} & 2.72295\times10^{-9} & 2.21275\times10^{-9} \end{pmatrix}$$

In[2537]:=

```
HueMtDistTo[{κmax, σmax, ϕmax}, {κ0, σ0, ϕ0}]
```

Out[2537]=

In[2538]:=

```
HueMtDistTo[{κ0, σ0, ϕ0}, {κ0, σ0, ϕ0}]
```

Out[2538]=

Examples

Note: Only the MTs on the sides of the box are plotted.
So you will only see the target point (blue) if it is on the side of the box.
Same for the max-distance (red) point.

plotting module

In[2539]:=

```
PlotBox[{κ0_, σ0_, ϕ0_}] := Module[{},
  Print["{κ₀, σ₀, θ₀} = ", N[{κ0, σ0, ϕ0} / Degree]];
  Print["{κmax, σmax, θmax} = ", MaxDist[{κ0, σ0, ϕ0}] / Degree];
  Print[Show[FoldUpBox[{κ0, σ0, ϕ0}], ImageSize → 350]];
  GraphicsGrid[{{BoxIn3D[{κ0, σ0, ϕ0}, "WantTop", "WantFront", "WantRight"],
     BoxIn3D[{κ0, σ0, ϕ0}, "noWantTop", "WantFront", "WantRight"]}}, ImageSize → 600]]
```

Horizontal fault (top of box) [dip $\theta = 0$]
Same view as Figure B2 of Tape and Tape (2012), “A geometric setting for moment tensors”.
(You need to wrap/close the top of the box to see that the blue segment agrees with the beachball positions.)

In[2540]:=

```
PlotBox[{π/3 + π/2, 0, 0}]
```

$\{\kappa_0, \sigma_0, \theta_0\} = \{150., 0., 0.\}$

$\{\kappa_{max}, \sigma_{max}, \theta_{max}\} = \{6.14052, 36.1405, 1.06769 \times 10^{-8}\}$



Out[2540]=



Horizontal fault (top of box) [dip $\theta = 0$]

In[2541]:=

```
PlotBox[{π/3, π/3, 0}]
```

$\{\kappa_0, \sigma_0, \theta_0\} = \{60., 60., 0.\}$

$\{\kappa_{max}, \sigma_{max}, \theta_{max}\} = \{179.665, -0.335058, 1.31074 \times 10^{-6}\}$



Out[2541]=



Normal dip-slip faults (front of box) [rake $\sigma = -90$]

In[2542]:=

```
PlotBox[{π/3, -π/2, π/3}]
```

$\{\kappa_0, \sigma_0, \theta_0\} = \{60., -90., 60.\}$

$\{\kappa_{max}, \sigma_{max}, \theta_{max}\} = \{60., 90., 60.\}$



Out[2542]=



Reverse dip-slip faults (back of box) -- same as Fig. 19 of Tape and Tape (2012) [rake σ = 90]

In[2543]:=

$$\text{PlotBox}\left[\left\{\frac{\pi}{3},\ \frac{\pi}{2},\ \frac{\pi}{3}\right\}\right]$$

$\{\kappa_0,\ \sigma_0,\ \theta_0\}$ = {60., 90., 60.}

$\{\kappa_{max},\ \sigma_{max},\ \theta_{max}\}$ = {60., -90., 60.}



Out[2543]=



Vertical faults (bottom of box) [dip $\theta = 90$]

In[2544]:=

```
PlotBox[{π/3, π/3, π/2}]
```

$\{\kappa_0, \sigma_0, \theta_0\} = \{60., 60., 90.\}$

$\{\kappa_{max}, \sigma_{max}, \theta_{max}\} = \{330., -6.90332 \times 10^{-9}, 30.\}$



Out[2544]=



Faults striking north (left AND right sides of box) [strike $\kappa = 0$]

In[2545]:=

```
PlotBox[{0, π/3, π/3}]
```

$\{\kappa_0, \sigma_0, \theta_0\} = \{0., 60., 60.\}$

$\{\kappa_{max}, \sigma_{max}, \theta_{max}\} = \{229.107, -49.1066, 41.4096\}$



Out[2545]=



Generic fault, interior to the strike-dip-rake box [all angles 60 degrees].
There are no dark blue regions, because there are no points on the box sides that are close to the target MT.

In[2546]:=

$$\texttt{PlotBox}\left[\left\{\frac{\pi}{3}, \frac{\pi}{3}, \frac{\pi}{3}\right\}\right]$$

$\{\kappa_0, \sigma_0, \theta_0\} = \{60., 60., 60.\}$

$\{\kappa_{max}, \sigma_{max}, \theta_{max}\} = \{289.107, -49.1066, 41.4096\}$



Out[2546]=